31/replication_package/data and codes/Descriptives/desc_composer_tables.do

args teacherstudent composers

////////////////////////////////////////////////////////////
// tables of composer level variables

use "`composers'", clear

// generate century of birth
gen century=floor(yr_born/100)+1
label define lbl_century 14 "14th" ///
15 "15th" 16 "16th" 17 "17th" ///
18 "18th" 19 "19th" 20 "20th"
label values century lbl_century

// tab birth_ctry, miss
preserve
egen composerid=group(composer)
collapse (count) composers=composerid, by(birth_ctry)
label var birth_ctry "birth_ctry"
label var composers "Composers"
export excel using "$tabs/descriptives.xlsx", ///
sheet(data_country) firstrow(varlab) sheetreplace
restore

// tab century, miss
preserve
egen composerid=group(composer)
collapse (count) composers=composerid, by(century)
label var century "Birth century"
label var composers "Composers"
export excel using "$tabs/descriptives.xlsx", ///
sheet(data_century) firstrow(varlab) sheetreplace
restore

// tab birth_ctry century, miss
preserve
egen composerid=group(composer)
collapse (count) composers=composerid, by(century birth_ctry)
reshape wide composers, i(birth_ctry) j(century)
foreach v of varlist composers* {
  replace `v'=0 if `v'==.
  local vLabel=substr("`v'",10,2)+"th"
  label var `v' "`vLabel'"
}
export excel using "$tabs/descriptives.xlsx", ///
sheet(data_country_century) firstrow(varlab) sheetreplace
restore

// tab teachers, miss
preserve
egen composerid=group(composer)
collapse (count) composers=composerid, by(teacher_ct)
label var teacher_ct "No. of teachers"
label var composers "Composers"
export excel using "$tabs/descriptives.xlsx", ///
sheet(data_teachers) firstrow(varlab) sheetreplace
restore

// tab students, miss
preserve
egen composerid=group(composer)
collapse (count) composers=composerid, by(student_ct)
label var student_ct "No. of students"
label var composers "Composers"
export excel using "$tabs/descriptives.xlsx", ///
sheet(data_students) firstrow(varlab) sheetreplace
restore

// tab BM teachers, miss
preserve
egen composerid=group(composer)
collapse (count) composers=composerid, by(bm_teacher_ct)
label var bm_teacher_ct "No. of BM teachers"
label var composers "Composers"
export excel using "$tabs/descriptives.xlsx", ///
sheet(data_bm_teachers) firstrow(varlab) sheetreplace
restore

// tab BM students, miss
preserve
egen composerid=group(composer)
collapse (count) composers=composerid, by(bm_student_ct)
label var bm_student_ct "No. of BM students"
label var composers "Composers"
export excel using "$tabs/descriptives.xlsx", ///
sheet(data_bm_students) firstrow(varlab) sheetreplace
restore